

```

# =====
# Compute the generalised connection, for later use in series expansion of  $x^a(s)$ 
# =====

::KeepHistory(false).
::PostDefaultRules( @@collect_terms!(%), @@sumflatten!(%) ).

{a,b,c,d,e,f,g,h,i,j,k,l,m,n,o,p,q,r,s,t,u#,v#}::Indices.

\nabla_{\#}::PartialDerivative.
\partial_{\#}::PartialDerivative.

g_{a b}::Metric.
g^{a b}::Metric.
\delta^a_b::KroneckerDelta.

R_{a b c d}::RiemannTensor.
R^a_{b c d}::RiemannTensor.

# --- the genQ are shorthand for the genGamma, saves typing -----
{genQ^a_{b c},genQ^a_{b c d},genQ^a_{b c d e},genQ^a_{b c d e f},Dx^a}::SortOrder.

genQ^a_{b c}::TableauSymmetry(shape={2}, indices={1,2}).
genQ^a_{b c d}::TableauSymmetry(shape={3}, indices={1,2,3}).
genQ^a_{b c d e}::TableauSymmetry(shape={4}, indices={1,2,3,4}).
genQ^a_{b c d e f}::TableauSymmetry(shape={5}, indices={1,2,3,4,5}).
genQ^a_{b c d e f g}::TableauSymmetry(shape={6}, indices={1,2,3,4,5,6}).

\Gamma^a_{b c}::TableauSymmetry(shape={2}, indices={1,2}).

# =====
# recursively compute  $d^n x^a/ds^n$  at  $s=0$ 
# =====

# ---  $A^a = x^a$ ,  $B^a = dx^a/ds$  at  $s=0$ 

deriv00:=A^a:
deriv01:=B^a:

deriv02:=-\Gamma^a_{b c} B^b B^c:

```

```

deriv03:=\nabla{@(deriv02)}:
@prodrule! (%):
@substitute! (%) (\nabla{B^{a}}->@(deriv02)):
@substitute! (%) (\nabla{\Gamma^m_{s t}}->B^d\partial_d{\Gamma^m_{s t}}):
@prodsort! (%): @rename_dummies! (%): @canonicalise! (%):

deriv04:=\nabla{@(deriv03)}:
@distribute! (%):
@prodrule! (%):
@substitute! (%) (\nabla{B^{a}}->@(deriv02)):
@substitute! (%) (\nabla{\Gamma^m_{s t}}->B^d\partial_d{\Gamma^m_{s t}}):
@substitute! (%) (\nabla{\partial_e{\Gamma^m_{s t}}}->B^d\partial_d{\partial_e{\Gamma^m_{s t}}}):
@prodsort! (%): @rename_dummies! (%): @canonicalise! (%):

deriv05:=\nabla{@(deriv04)}:
@distribute! (%):
@prodrule! (%):
@substitute! (%) (\nabla{B^{a}}->@(deriv02)):
@substitute! (%) (\nabla{\Gamma^m_{s t}}->B^d\partial_d{\Gamma^m_{s t}}):
@substitute! (%) (\nabla{\partial_e{\Gamma^m_{s t}}}->B^d\partial_d{\partial_e{\Gamma^m_{s t}}}):
@substitute! (%) (\nabla{\partial_{ef}{\Gamma^m_{s t}}}->B^d\partial_d{\partial_{ef}{\Gamma^m_{s t}}}):
@prodsort! (%): @rename_dummies! (%): @canonicalise! (%):

# --- deriv06 to leading order in s equals  $-\Gamma^a_{bcdefg}$  which vanishes only in RNC

deriv06:=\nabla{@(deriv05)}:
@distribute! (%):
@prodrule! (%):
@substitute! (%) (\nabla{B^{a}}->@(deriv02)):
@substitute! (%) (\nabla{\Gamma^m_{s t}}->B^d\partial_d{\Gamma^m_{s t}}):
@substitute! (%) (\nabla{\partial_e{\Gamma^m_{s t}}}->B^d\partial_d{\partial_e{\Gamma^m_{s t}}}):
@substitute! (%) (\nabla{\partial_{ef}{\Gamma^m_{s t}}}->B^d\partial_d{\partial_{ef}{\Gamma^m_{s t}}}):
@substitute! (%) (\nabla{\partial_{efg}{\Gamma^m_{s t}}}->B^d\partial_d{\partial_{efg}{\Gamma^m_{s t}}}):
@prodsort! (%): @rename_dummies! (%): @canonicalise! (%):

genGamma02:= - @(deriv02): @substitute! (%) (B^a -> 1):
genGamma03:= - @(deriv03): @substitute! (%) (B^a -> 1):
genGamma04:= - @(deriv04): @substitute! (%) (B^a -> 1):

```

```

genGamma05:= - @(deriv05)): @substitute!(%)(B^a -> 1):
genGamma06:= - @(deriv06)): @substitute!(%)(B^a -> 1):

@print["\Btag{01}\genGammaA=~@(genGamma02)~"\Etag{01}"];
@print["\Btag{02}\genGammaB=\killL"~@(genGamma03)~"\killR\Etag{02}"];
@print["\Btag{03}\genGammaC=\killL"~@(genGamma04)~"\killR\Etag{03}"];
@print["\Btag{04}\genGammaD=\killL"~@(genGamma05)~"\killR\Etag{04}"];
@print["\Btag{05}\genGammaE=\killL"~@(genGamma06)~"\killR\Etag{05}"];

```

# --- the generalised connections in arbitrary coordinates -----

$$\Gamma_{(bc)}^a(x) = \Gamma^a{}_{bc}$$

$$\Gamma_{(bcd)}^a(x) = \partial_b \Gamma^a{}_{cd} - 2 \Gamma^a{}_{be} \Gamma^e{}_{cd}$$

$$\Gamma_{(bcde)}^a(x) = -\Gamma^f{}_{bc} \partial_f \Gamma^a{}_{de} - 4 \Gamma^f{}_{bc} \partial_d \Gamma^a{}_{ef} + \partial_{bc} \Gamma^a{}_{de} + 2 \Gamma^a{}_{fg} \Gamma^f{}_{bc} \Gamma^g{}_{de} + 4 \Gamma^a{}_{bf} \Gamma^f{}_{cg} \Gamma^g{}_{de} - 2 \Gamma^a{}_{bf} \partial_c \Gamma^f{}_{de}$$

$$\begin{aligned} \Gamma_{(bcdef)}^a(x) = & 2 \Gamma^g{}_{bc} \Gamma^h{}_{dg} \partial_h \Gamma^a{}_{ef} + 6 \Gamma^g{}_{bc} \Gamma^h{}_{de} \partial_g \Gamma^a{}_{fh} - \partial_g \Gamma^a{}_{bc} \partial_d \Gamma^g{}_{ef} - 3 \Gamma^g{}_{bc} \partial_{dg} \Gamma^a{}_{ef} + 12 \Gamma^g{}_{bc} \Gamma^h{}_{dg} \partial_e \Gamma^a{}_{fh} + 6 \Gamma^g{}_{bc} \Gamma^h{}_{de} \partial_f \Gamma^a{}_{gh} \\ & - 6 \partial_b \Gamma^a{}_{cg} \partial_d \Gamma^g{}_{ef} - 6 \Gamma^g{}_{bc} \partial_{de} \Gamma^a{}_{fg} + \partial_{bcd} \Gamma^a{}_{ef} - 12 \Gamma^a{}_{gh} \Gamma^g{}_{bc} \Gamma^h{}_{di} \Gamma^i{}_{ef} + 6 \Gamma^a{}_{gh} \Gamma^g{}_{bc} \partial_d \Gamma^h{}_{ef} - 4 \Gamma^a{}_{bg} \Gamma^g{}_{hi} \Gamma^h{}_{cd} \Gamma^i{}_{ef} \\ & - 8 \Gamma^a{}_{bg} \Gamma^g{}_{ch} \Gamma^h{}_{di} \Gamma^i{}_{ef} + 8 \Gamma^a{}_{bg} \Gamma^h{}_{cd} \partial_e \Gamma^g{}_{fh} + 4 \Gamma^a{}_{bg} \Gamma^g{}_{ch} \partial_d \Gamma^h{}_{ef} + 2 \Gamma^a{}_{bg} \Gamma^h{}_{cd} \partial_h \Gamma^g{}_{ef} - 2 \Gamma^a{}_{bg} \partial_{cd} \Gamma^g{}_{ef} \end{aligned}$$

$$\begin{aligned} \Gamma_{(bcdefg)}^a(x) = & -4 \Gamma^h{}_{bc} \Gamma^i{}_{dh} \Gamma^j{}_{ei} \partial_j \Gamma^a{}_{fg} - 2 \Gamma^h{}_{bc} \Gamma^i{}_{de} \Gamma^j{}_{hi} \partial_j \Gamma^a{}_{fg} - 16 \Gamma^h{}_{bc} \Gamma^i{}_{de} \Gamma^j{}_{fh} \partial_j \Gamma^a{}_{gi} + 2 \Gamma^h{}_{bi} \partial_h \Gamma^a{}_{cd} \partial_e \Gamma^i{}_{fg} + 4 \Gamma^h{}_{bc} \partial_i \Gamma^a{}_{de} \partial_f \Gamma^i{}_{gh} + 8 \Gamma^h{}_{bc} \Gamma^i{}_{dh} \partial_{ei} \Gamma^a{}_{fg} \\ & - 24 \Gamma^h{}_{bc} \Gamma^i{}_{de} \Gamma^j{}_{fh} \partial_i \Gamma^a{}_{gj} - 12 \Gamma^h{}_{bc} \Gamma^i{}_{de} \Gamma^j{}_{fg} \partial_h \Gamma^a{}_{ij} + 8 \Gamma^h{}_{bc} \partial_i \Gamma^a{}_{dh} \partial_e \Gamma^i{}_{fg} + 12 \Gamma^h{}_{bc} \partial_h \Gamma^a{}_{di} \partial_e \Gamma^i{}_{fg} + 24 \Gamma^h{}_{bc} \Gamma^i{}_{de} \partial_{fh} \Gamma^a{}_{gi} \\ & + \Gamma^h{}_{bc} \partial_i \Gamma^a{}_{de} \partial_h \Gamma^i{}_{fg} - 4 \partial_b \Gamma^h{}_{cd} \partial_{eh} \Gamma^a{}_{fg} - \partial_h \Gamma^a{}_{bc} \partial_{de} \Gamma^h{}_{fg} + 3 \Gamma^h{}_{bc} \Gamma^i{}_{de} \partial_{hi} \Gamma^a{}_{fg} - 6 \Gamma^h{}_{bc} \partial_{deh} \Gamma^a{}_{fg} - 32 \Gamma^h{}_{bc} \Gamma^i{}_{dh} \Gamma^j{}_{ei} \partial_f \Gamma^a{}_{gj} \\ & - 16 \Gamma^h{}_{bc} \Gamma^i{}_{de} \Gamma^j{}_{hi} \partial_f \Gamma^a{}_{gj} - 48 \Gamma^h{}_{bc} \Gamma^i{}_{de} \Gamma^j{}_{fh} \partial_g \Gamma^a{}_{ij} + 16 \Gamma^h{}_{bi} \partial_c \Gamma^a{}_{dh} \partial_e \Gamma^i{}_{fg} + 32 \Gamma^h{}_{bc} \partial_d \Gamma^a{}_{ei} \partial_f \Gamma^i{}_{gh} + 24 \Gamma^h{}_{bc} \Gamma^i{}_{dh} \partial_{ef} \Gamma^a{}_{gi} \\ & + 24 \Gamma^h{}_{bc} \partial_d \Gamma^a{}_{hi} \partial_e \Gamma^i{}_{fg} + 12 \Gamma^h{}_{bc} \Gamma^i{}_{de} \partial_{fg} \Gamma^a{}_{hi} + 8 \Gamma^h{}_{bc} \partial_d \Gamma^a{}_{ei} \partial_h \Gamma^i{}_{fg} - 12 \partial_b \Gamma^h{}_{cd} \partial_{ef} \Gamma^a{}_{gh} - 8 \partial_b \Gamma^a{}_{ch} \partial_{de} \Gamma^h{}_{fg} - 8 \Gamma^h{}_{bc} \partial_{def} \Gamma^a{}_{gh} + \partial_{bcde} \Gamma^a{}_{fg} \\ & + 24 \Gamma^a{}_{hi} \Gamma^h{}_{bj} \Gamma^i{}_{ck} \Gamma^j{}_{de} \Gamma^k{}_{fg} + 16 \Gamma^a{}_{hi} \Gamma^h{}_{bc} \Gamma^i{}_{jk} \Gamma^j{}_{de} \Gamma^k{}_{fg} + 32 \Gamma^a{}_{hi} \Gamma^h{}_{bc} \Gamma^i{}_{dj} \Gamma^j{}_{ek} \Gamma^k{}_{fg} - 24 \Gamma^a{}_{hi} \Gamma^h{}_{bj} \Gamma^j{}_{cd} \partial_e \Gamma^i{}_{fg} - 32 \Gamma^a{}_{hi} \Gamma^h{}_{bc} \Gamma^j{}_{de} \partial_f \Gamma^i{}_{gj} \\ & - 16 \Gamma^a{}_{hi} \Gamma^h{}_{bc} \Gamma^i{}_{dj} \partial_e \Gamma^j{}_{fg} - 8 \Gamma^a{}_{hi} \Gamma^h{}_{bc} \Gamma^j{}_{de} \partial_j \Gamma^i{}_{fg} + 6 \Gamma^a{}_{hi} \partial_b \Gamma^h{}_{cd} \partial_e \Gamma^i{}_{fg} + 8 \Gamma^a{}_{hi} \Gamma^h{}_{bc} \partial_{de} \Gamma^i{}_{fg} + 24 \Gamma^a{}_{bh} \Gamma^h{}_{ij} \Gamma^i{}_{cd} \Gamma^j{}_{ek} \Gamma^k{}_{fg} \\ & - 12 \Gamma^a{}_{bh} \Gamma^i{}_{cd} \Gamma^j{}_{ef} \partial_g \Gamma^h{}_{ij} - 12 \Gamma^a{}_{bh} \Gamma^h{}_{ij} \Gamma^i{}_{cd} \partial_e \Gamma^j{}_{fg} + 8 \Gamma^a{}_{bh} \Gamma^h{}_{ci} \Gamma^i{}_{jk} \Gamma^j{}_{de} \Gamma^k{}_{fg} + 16 \Gamma^a{}_{bh} \Gamma^h{}_{ci} \Gamma^i{}_{dj} \Gamma^j{}_{ek} \Gamma^k{}_{fg} - 24 \Gamma^a{}_{bh} \Gamma^i{}_{cd} \Gamma^j{}_{ei} \partial_f \Gamma^h{}_{gj} \\ & - 16 \Gamma^a{}_{bh} \Gamma^h{}_{ci} \Gamma^j{}_{de} \partial_f \Gamma^i{}_{gj} - 8 \Gamma^a{}_{bh} \Gamma^h{}_{ci} \Gamma^i{}_{dj} \partial_e \Gamma^j{}_{fg} - 12 \Gamma^a{}_{bh} \Gamma^i{}_{cd} \Gamma^j{}_{ef} \partial_i \Gamma^h{}_{gj} + 12 \Gamma^a{}_{bh} \partial_c \Gamma^h{}_{di} \partial_e \Gamma^i{}_{fg} + 12 \Gamma^a{}_{bh} \Gamma^i{}_{cd} \partial_{ef} \Gamma^h{}_{gi} \\ & - 4 \Gamma^a{}_{bh} \Gamma^h{}_{ci} \Gamma^j{}_{de} \partial_j \Gamma^i{}_{fg} + 4 \Gamma^a{}_{bh} \Gamma^h{}_{ci} \partial_{de} \Gamma^i{}_{fg} - 4 \Gamma^a{}_{bh} \Gamma^i{}_{cd} \Gamma^j{}_{ei} \partial_j \Gamma^h{}_{fg} + 2 \Gamma^a{}_{bh} \partial_c \Gamma^i{}_{de} \partial_i \Gamma^h{}_{fg} + 6 \Gamma^a{}_{bh} \Gamma^i{}_{cd} \partial_{ei} \Gamma^h{}_{fg} - 2 \Gamma^a{}_{bh} \partial_{cde} \Gamma^h{}_{fg} \end{aligned}$$

```

genQ02:=@(genGamma02): @sym!({_b,_c}:
genQ03:=@(genGamma03): @sym!({_b,_c,_d}:
genQ04:=@(genGamma04): @sym!({_b,_c,_d,_e}:
genQ05:=@(genGamma05): @sym!({_b,_c,_d,_e,_f}:

# --- compute coordinate transforms from any x^a to rnc y^a -----

y1:=Dx^a;
y2:=Dx^a + (1/2) genQ^{a}_{b c} y1^b y1^c;
y3:=Dx^a + (1/2) genQ^{a}_{b c} y2^b y2^c
+ (1/6) genQ^{a}_{b c d} y1^b y1^c y1^d;
y4:=Dx^a + (1/2) genQ^{a}_{b c} y3^b y3^c
+ (1/6) genQ^{a}_{b c d} y2^b y2^c y2^d
+ (1/24) genQ^{a}_{b c d e} y1^b y1^c y1^d y1^e;
y5:=Dx^a + (1/2) genQ^{a}_{b c} y4^b y4^c
+ (1/6) genQ^{a}_{b c d} y3^b y3^c y3^d
+ (1/24) genQ^{a}_{b c d e} y2^b y2^c y2^d y2^e
+ (1/120) genQ^{a}_{b c d e f} y1^b y1^c y1^d y1^e y1^f;

@substitute!(y2)(y1^a -> @(y1));

# --- truncate y3 to 2nd order in \eps -----

@substitute!(y3)(y2^a -> @(y2));
@substitute!(y3)(y1^a -> @(y1));
@distributed!(%):
@prodsort!(%):
@rename_dummies!(%):
@canonicalise!(%):

poly:=@(y3):

genQ^{a}_{b c}::Weight(label=Qterms,value=1).
genQ^{a}_{b c d}::Weight(label=Qterms,value=2).

term00:=@(poly): @keep_weight!(term00){Qterms}{0}:
term01:=@(poly): @keep_weight!(term01){Qterms}{1}:
term02:=@(poly): @keep_weight!(term02){Qterms}{2}:

y3:=@(term00) + @(term01) + @(term02):

```

```

# --- tidy up -----
@prodsort!(%):
@rename_dummies!(%):
@canonicalise!(%);

# --- truncate y4 to 3rd order in \eps -----
@substitute!(y4)(y3^a -> @(y3)):
@substitute!(y4)(y2^a -> @(y2)):
@substitute!(y4)(y1^a -> @(y1)):
@distribute!(%):
@prodsort!(%):
@rename_dummies!(%):
@canonicalise!(%):

poly:=@(y4):

genQ^{a}_{b c}::Weight(label=Qterms,value=1).
genQ^{a}_{b c d}::Weight(label=Qterms,value=2).
genQ^{a}_{b c d e}::Weight(label=Qterms,value=3).

term00:=@(poly): @keep_weight!(term00){Qterms}{0}:
term01:=@(poly): @keep_weight!(term01){Qterms}{1}:
term02:=@(poly): @keep_weight!(term02){Qterms}{2}:
term03:=@(poly): @keep_weight!(term03){Qterms}{3}:

y4:=@(term00) + @(term01) + @(term02) + @(term03):

# --- tidy up -----
@prodsort!(%):
@rename_dummies!(%):
@canonicalise!(%);

# --- truncate y5 to 4th order in \eps -----
@substitute!(y5)(y4^a -> @(y4)):
@substitute!(y5)(y3^a -> @(y3)):
@substitute!(y5)(y2^a -> @(y2)):
@substitute!(y5)(y1^a -> @(y1)):

```

```

@distribute! (%):
@prodsort! (%):
@rename_dummies! (%):
@canonicalise! (%):

poly:=@(y5):

genQ^{a}_{b c}::Weight(label=Qterms,value=1).
genQ^{a}_{b c d}::Weight(label=Qterms,value=2).
genQ^{a}_{b c d e}::Weight(label=Qterms,value=3).
genQ^{a}_{b c d e f}::Weight(label=Qterms,value=4).

term00:=@(poly): @keep_weight!(term00){Qterms}{0}:
term01:=@(poly): @keep_weight!(term01){Qterms}{1}:
term02:=@(poly): @keep_weight!(term02){Qterms}{2}:
term03:=@(poly): @keep_weight!(term03){Qterms}{3}:
term04:=@(poly): @keep_weight!(term04){Qterms}{4}:

y5:=@(term00) + @(term01) + @(term02) + @(term03) + @(term04):

# --- tidy up -----
@prodsort! (%):
@rename_dummies! (%):
@canonicalise! (%);

```

$$y1 := Dx^a$$

$$y2 := Dx^a + \frac{1}{2} genQ^a_{bc} y1^b y1^c$$

$$y3 := Dx^a + \frac{1}{2} genQ^a_{bc} y2^b y2^c + \frac{1}{6} genQ^a_{bcd} y1^b y1^c y1^d$$

$$y4 := Dx^a + \frac{1}{2} genQ^a_{bc} y3^b y3^c + \frac{1}{6} genQ^a_{bcd} y2^b y2^c y2^d + \frac{1}{24} genQ^a_{bcde} y1^b y1^c y1^d y1^e$$

$$y5 := Dx^a + \frac{1}{2} genQ^a_{bc} y4^b y4^c + \frac{1}{6} genQ^a_{bcd} y3^b y3^c y3^d + \frac{1}{24} genQ^a_{bcde} y2^b y2^c y2^d y2^e + \frac{1}{120} genQ^a_{bcdef} y1^b y1^c y1^d y1^e y1^f$$

$$y2 := Dx^a + \frac{1}{2} genQ^a_{bc} Dx^b Dx^c$$

$$y3 := Dx^a + \frac{1}{2} genQ^a_{bc} Dx^b Dx^c + \frac{1}{2} genQ^a_{bc} genQ^b_{de} Dx^c Dx^d Dx^e + \frac{1}{6} genQ^a_{bcd} Dx^b Dx^c Dx^d$$

$$y4 := Dx^a + \frac{1}{2} genQ^a_{bc} Dx^b Dx^c + \frac{1}{2} genQ^a_{bc} genQ^b_{de} Dx^c Dx^d Dx^e + \frac{1}{6} genQ^a_{bcd} Dx^b Dx^c Dx^d + \frac{1}{2} genQ^a_{bc} genQ^b_{de} genQ^d_{fg} Dx^c Dx^e Dx^f Dx^g \\ + \frac{1}{6} genQ^a_{bc} genQ^b_{def} Dx^c Dx^d Dx^e Dx^f + \frac{1}{8} genQ^a_{bc} genQ^b_{de} genQ^c_{fg} Dx^d Dx^e Dx^f Dx^g + \frac{1}{4} genQ^b_{cd} genQ^a_{bef} Dx^c Dx^d Dx^e Dx^f + \frac{1}{24} genQ^a_{bcde} Dx^b Dx^c Dx^d Dx^e$$

$$y5 := Dx^a + \frac{1}{2} genQ^a_{bc} Dx^b Dx^c + \frac{1}{2} genQ^a_{bc} genQ^b_{de} Dx^c Dx^d Dx^e + \frac{1}{6} genQ^a_{bcd} Dx^b Dx^c Dx^d + \frac{1}{2} genQ^a_{bc} genQ^b_{de} genQ^d_{fg} Dx^c Dx^e Dx^f Dx^g \\ + \frac{1}{6} genQ^a_{bc} genQ^b_{def} Dx^c Dx^d Dx^e Dx^f + \frac{1}{8} genQ^a_{bc} genQ^b_{de} genQ^c_{fg} Dx^d Dx^e Dx^f Dx^g + \frac{1}{4} genQ^b_{cd} genQ^a_{bef} Dx^c Dx^d Dx^e Dx^f \\ + \frac{1}{24} genQ^a_{bcde} Dx^b Dx^c Dx^d Dx^e + \frac{1}{2} genQ^a_{bc} genQ^b_{de} genQ^d_{fg} genQ^f_{hi} Dx^c Dx^e Dx^g Dx^h Dx^i + \frac{1}{6} genQ^a_{bc} genQ^b_{de} genQ^d_{fgh} Dx^c Dx^e Dx^f Dx^g Dx^h \\ + \frac{1}{8} genQ^a_{bc} genQ^b_{de} genQ^d_{fg} genQ^e_{hi} Dx^c Dx^f Dx^g Dx^h Dx^i + \frac{1}{4} genQ^a_{bc} genQ^d_{ef} genQ^b_{dgh} Dx^c Dx^e Dx^f Dx^g Dx^h \\ + \frac{1}{24} genQ^a_{bc} genQ^b_{defg} Dx^c Dx^d Dx^e Dx^f Dx^g + \frac{1}{4} genQ^a_{bc} genQ^b_{de} genQ^c_{fg} genQ^d_{hi} Dx^e Dx^f Dx^g Dx^h Dx^i \\ + \frac{1}{12} genQ^a_{bc} genQ^b_{de} genQ^c_{fgh} Dx^d Dx^e Dx^f Dx^g Dx^h + \frac{1}{4} genQ^b_{cd} genQ^c_{ef} genQ^a_{bgh} Dx^d Dx^e Dx^f Dx^g Dx^h + \frac{1}{12} genQ^a_{bcd} genQ^b_{efg} Dx^c Dx^d Dx^e Dx^f Dx^g \\ + \frac{1}{8} genQ^b_{cd} genQ^e_{fg} genQ^a_{beh} Dx^c Dx^d Dx^f Dx^g Dx^h + \frac{1}{12} genQ^b_{cd} genQ^a_{befg} Dx^c Dx^d Dx^e Dx^f Dx^g + \frac{1}{120} genQ^a_{bcdef} Dx^b Dx^c Dx^d Dx^e Dx^f$$



```

# --- y^a in terms of \Gamma and its derivatives -----

@substitute!(y2)(genQ^{a}_{b c} -> @(genQ02)):
@distributed!(%):
@prodsort!(%):
@rename_dummies!(%):
@canonicalise!(%):

@substitute!(y3)(genQ^{a}_{b c} -> @(genQ02)):
@substitute!(y3)(genQ^{a}_{b c d} -> @(genQ03)):
@distributed!(%):
@prodsort!(%):
@rename_dummies!(%):
@canonicalise!(%):

@substitute!(y4)(genQ^{a}_{b c} -> @(genQ02)):
@substitute!(y4)(genQ^{a}_{b c d} -> @(genQ03)):
@substitute!(y4)(genQ^{a}_{b c d e} -> @(genQ04)):
@distributed!(%):
@prodsort!(%):
@rename_dummies!(%):
@canonicalise!(%):

@substitute!(y5)(genQ^{a}_{b c} -> @(genQ02)):
@substitute!(y5)(genQ^{a}_{b c d} -> @(genQ03)):
@substitute!(y5)(genQ^{a}_{b c d e} -> @(genQ04)):
@substitute!(y5)(genQ^{a}_{b c d e f} -> @(genQ05)):
@distributed!(%):
@prodsort!(%):
@rename_dummies!(%):
@canonicalise!(%):

tmp00:=@(y1):
@substitute!(tmp00)(Dx^{a} -> \Delta{x}^{a}):

tmp01:=@(y2)-@(y1):
@factor_out!!(tmp01)(Dx^{a}):
@substitute!(tmp01)(Dx^{a} -> \Delta{x}^{a}):

```

```
tmp02:=@(y3)-@(y2):
@factor_out!!(tmp02)(Dx^{a}):
@substitute!(tmp02)(Dx^{a} -> \Delta{x}^{a}):
```

```
tmp03:=@(y4)-@(y3):
@factor_out!!(tmp03)(Dx^{a}):
@substitute!(tmp03)(Dx^{a} -> \Delta{x}^{a}):
```

```
tmp04:=@(y5)-@(y4):
@factor_out!!(tmp04)(Dx^{a}):
@substitute!(tmp04)(Dx^{a} -> \Delta{x}^{a}):
```

```
genx2rnc:=@(tmp00)+@(tmp01)+@(tmp02)+@(tmp03)+@(tmp04); "genx2rnc.trn"
```

```
# --- y^a in terms of \Gamma and its derivatives -----
```

```
@print["\Btag{20}y^a_5="~@(genx2rnc)~"\Etag{20}"];
```

$$\begin{aligned}
y_5^a = & \left( \Delta x^a + \frac{1}{2} \Delta x^b \Delta x^c \Gamma^a_{bc} + \Delta x^b \Delta x^c \Delta x^d \left( \frac{1}{6} \Gamma^a_{be} \Gamma^e_{cd} + \frac{1}{6} \partial_b \Gamma^a_{cd} \right) + \Delta x^b \Delta x^c \Delta x^d \Delta x^e \left( \frac{1}{12} \Gamma^a_{bf} \partial_c \Gamma^f_{de} + \frac{1}{24} \Gamma^a_{fg} \Gamma^f_{bc} \Gamma^g_{de} + \frac{1}{24} \Gamma^f_{bc} \partial_f \Gamma^a_{de} + \frac{1}{24} \partial_{bc} \Gamma^a_{de} \right) \right. \\
& + \Delta x^b \Delta x^c \Delta x^d \Delta x^e \Delta x^f \left( -\frac{1}{90} \Gamma^a_{bg} \Gamma^g_{ch} \Gamma^h_{di} \Gamma^i_{ef} + \frac{1}{180} \Gamma^a_{bg} \Gamma^g_{ch} \partial_d \Gamma^h_{ef} + \frac{1}{120} \Gamma^a_{bg} \Gamma^g_{hi} \Gamma^h_{cd} \Gamma^i_{ef} + \frac{1}{60} \Gamma^a_{bg} \Gamma^h_{cd} \partial_h \Gamma^g_{ef} - \frac{1}{60} \Gamma^a_{bg} \Gamma^h_{cd} \partial_e \Gamma^g_{fh} \right. \\
& + \frac{1}{40} \Gamma^a_{bg} \partial_{cd} \Gamma^g_{ef} + \frac{1}{90} \Gamma^a_{gh} \Gamma^g_{bc} \Gamma^h_{di} \Gamma^i_{ef} + \frac{13}{360} \Gamma^a_{gh} \Gamma^g_{bc} \partial_d \Gamma^h_{ef} + \frac{1}{360} \Gamma^g_{bc} \Gamma^h_{dg} \partial_h \Gamma^a_{ef} - \frac{1}{90} \Gamma^g_{bc} \Gamma^h_{dg} \partial_e \Gamma^a_{fh} + \frac{7}{360} \partial_g \Gamma^a_{bc} \partial_d \Gamma^g_{ef} + \frac{1}{180} \partial_b \Gamma^a_{cg} \partial_d \Gamma^g_{ef} \\
& \left. \left. + \frac{1}{120} \Gamma^g_{bc} \Gamma^h_{de} \partial_g \Gamma^a_{fh} + \frac{1}{120} \Gamma^g_{bc} \Gamma^h_{de} \partial_f \Gamma^a_{gh} + \frac{1}{60} \Gamma^g_{bc} \partial_{dg} \Gamma^a_{ef} - \frac{1}{120} \Gamma^g_{bc} \partial_{de} \Gamma^a_{fg} + \frac{1}{120} \partial_{bcd} \Gamma^a_{ef} \right) \right)
\end{aligned}$$

```

# =====
# From here on we are adapting the genGamma's to RNC
# =====
#
# Now we need to expand deriv0* as a Taylor series around x=0
# These will be used in constructing the general solution of the geodesic equation near x=0, i.e. x^a(s)
# To obtain a series accurate to terms including O(L^6) we will need the following derivatives at x=0
#
#   deriv02,a   deriv02,ab   deriv02,abc   deriv02,abcd
#   deriv03,a   deriv03,ab   deriv03,abc
#   deriv04,a   deriv04,ab
#   deriv05,a
#
# The Q^{a}_{b c d e f\cdots} are the partial derivatives of \Gamma^a_{b c}
#
# =====

# --- Taylor series for deriv02 -----

@distributed!(deriv02):
term00:=@(deriv02):

tmp:=A^{a} \partial_{a}{@(term00)}:
@distributed!(tmp): @prodrule!(tmp): @distributed!(tmp):
@substitute!(tmp)(\partial_{a}{B^{b}} -> 0):
term01:=@(tmp):

tmp:=A^{a} \partial_{a}{@(term01)}:
@distributed!(tmp): @prodrule!(tmp): @distributed!(tmp):
@substitute!(tmp)(\partial_{a}{B^{b}} -> 0, \partial_{a}{A^{b}} -> 0):
term02:=@(tmp):

tmp:=A^{a} \partial_{a}{@(term02)}:
@distributed!(tmp): @prodrule!(tmp): @distributed!(tmp):
@substitute!(tmp)(\partial_{a}{B^{b}} -> 0, \partial_{a}{A^{b}} -> 0):
term03:=@(tmp):

tmp:=A^{a} \partial_{a}{@(term03)}:
@distributed!(tmp): @prodrule!(tmp): @distributed!(tmp):

```

```

@substitute!(tmp)(\partial_{a}{B^{b}} -> 0, \partial_{a}{A^{b}} -> 0):
term04:=@(tmp):

@substitute!(term00)(\partial_{a b c d}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b c d}):
@substitute!(term00)(\partial_{a b c}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b c}):
@substitute!(term00)(\partial_{a b}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b}):
@substitute!(term00)(\partial_{a}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a}):
@substitute!(term00)(\Gamma^{p}_{m n} -> Q^{p}_{m n}):

@substitute!(term01)(\partial_{a b c d}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b c d}):
@substitute!(term01)(\partial_{a b c}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b c}):
@substitute!(term01)(\partial_{a b}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b}):
@substitute!(term01)(\partial_{a}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a}):
@substitute!(term01)(\Gamma^{p}_{m n} -> Q^{p}_{m n}):

@substitute!(term02)(\partial_{a b c d}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b c d}):
@substitute!(term02)(\partial_{a b c}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b c}):
@substitute!(term02)(\partial_{a b}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b}):
@substitute!(term02)(\partial_{a}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a}):
@substitute!(term02)(\Gamma^{p}_{m n} -> Q^{p}_{m n}):

@substitute!(term03)(\partial_{a b c d}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b c d}):
@substitute!(term03)(\partial_{a b c}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b c}):
@substitute!(term03)(\partial_{a b}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b}):
@substitute!(term03)(\partial_{a}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a}):
@substitute!(term03)(\Gamma^{p}_{m n} -> Q^{p}_{m n}):

@substitute!(term04)(\partial_{a b c d}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b c d}):
@substitute!(term04)(\partial_{a b c}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b c}):
@substitute!(term04)(\partial_{a b}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b}):
@substitute!(term04)(\partial_{a}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a}):
@substitute!(term04)(\Gamma^{p}_{m n} -> Q^{p}_{m n}):

# --- rebuild deriv02
deriv02:=@(term00) + @(term01) + (1/2) @(term02) + (1/6) @(term03) + (1/24) @(term04):

# --- impose RNC conditions
@substitute!(deriv02)(Q^{p}_{m n} -> 0):

```

```

@substitute!(deriv02)(B^a B^b Q^{p}_{a b} -> 0);

# --- Taylor series for deriv03 -----

@distributed!(deriv03):
term00:=@deriv03:

tmp:=A^{a} \partial_{a}{@(term00)}:
@distributed!(tmp): @prodrule!(tmp): @distributed!(tmp):
@substitute!(tmp)(\partial_{a}{B^{b}} -> 0):
term01:=@tmp:

tmp:=A^{a} \partial_{a}{@(term01)}:
@distributed!(tmp): @prodrule!(tmp): @distributed!(tmp):
@substitute!(tmp)(\partial_{a}{B^{b}} -> 0, \partial_{a}{A^{b}} -> 0):
term02:=@tmp:

tmp:=A^{a} \partial_{a}{@(term02)}:
@distributed!(tmp): @prodrule!(tmp): @distributed!(tmp):
@substitute!(tmp)(\partial_{a}{B^{b}} -> 0, \partial_{a}{A^{b}} -> 0):
term03:=@tmp:

@substitute!(term00)(\partial_{a b c}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b c}):
@substitute!(term00)(\partial_{a b}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b}):
@substitute!(term00)(\partial_{a}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a}):
@substitute!(term00)(\Gamma^{p}_{m n} -> Q^{p}_{m n}):

@substitute!(term01)(\partial_{a b c}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b c}):
@substitute!(term01)(\partial_{a b}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b}):
@substitute!(term01)(\partial_{a}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a}):
@substitute!(term01)(\Gamma^{p}_{m n} -> Q^{p}_{m n}):

@substitute!(term02)(\partial_{a b c}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b c}):
@substitute!(term02)(\partial_{a b}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b}):
@substitute!(term02)(\partial_{a}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a}):
@substitute!(term02)(\Gamma^{p}_{m n} -> Q^{p}_{m n}):

@substitute!(term03)(\partial_{a b c}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b c}):
@substitute!(term03)(\partial_{a b}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b}):

```

```

@substitute!(term03)(\partial_{a}\{\Gamma^{p}_{m n}\} -> Q^{p}_{m n a}):
@substitute!(term03)(\Gamma^{p}_{m n} -> Q^{p}_{m n}):

# --- rebuild deriv03

deriv03:=@(term00) + @(term01) + (1/2) @(term02) + (1/6) @(term03):

# --- impose RNC conditions

@substitute!(deriv03)(Q^{p}_{m n} -> 0):
@substitute!(deriv03)(B^a B^b B^c Q^{p}_{a b c} -> 0);

# --- Taylor series for deriv04 -----

@distributed!(deriv04):
term00:=@(deriv04):

tmp:=A^{a} \partial_{a}\{@(term00)\}:
@distributed!(tmp): @prodrule!(tmp): @distributed!(tmp):
@substitute!(tmp)(\partial_{a}\{B^b\} -> 0):
term01:=@(tmp):

tmp:=A^{a} \partial_{a}\{@(term01)\}:
@distributed!(tmp): @prodrule!(tmp): @distributed!(tmp):
@substitute!(tmp)(\partial_{a}\{B^b\} -> 0, \partial_{a}\{A^b\} -> 0):
term02:=@(tmp):

@substitute!(term00)(\partial_{a b}\{\Gamma^{p}_{m n}\} -> Q^{p}_{m n a b}):
@substitute!(term00)(\partial_{a}\{\Gamma^{p}_{m n}\} -> Q^{p}_{m n a}):
@substitute!(term00)(\Gamma^{p}_{m n} -> Q^{p}_{m n}):

@substitute!(term01)(\partial_{a b}\{\Gamma^{p}_{m n}\} -> Q^{p}_{m n a b}):
@substitute!(term01)(\partial_{a}\{\Gamma^{p}_{m n}\} -> Q^{p}_{m n a}):
@substitute!(term01)(\Gamma^{p}_{m n} -> Q^{p}_{m n}):

@substitute!(term02)(\partial_{a b}\{\Gamma^{p}_{m n}\} -> Q^{p}_{m n a b}):
@substitute!(term02)(\partial_{a}\{\Gamma^{p}_{m n}\} -> Q^{p}_{m n a}):
@substitute!(term02)(\Gamma^{p}_{m n} -> Q^{p}_{m n}):

# --- rebuild deriv04

deriv04:=@(term00) + @(term01) + (1/2) @(term02):

```

```

# --- impose RNC conditions

@substitute!(deriv04)(Q^{p}_{m n} -> 0):
@substitute!(deriv04)(B^a B^b B^c Q^{p}_{a b c} -> 0):
@substitute!(deriv04)(B^a B^b B^c B^d Q^{p}_{a b c d} -> 0);

# --- Taylor series for deriv05 -----

@distributed!(deriv05):
term00:=@(deriv05):

tmp:=A^{a} \partial_{a}{@(term00)}:
@distributed!(tmp): @prodrule!(tmp): @distributed!(tmp):
@substitute!(tmp)(\partial_{a}{B^{b}} -> 0):
term01:=@(tmp):

@substitute!(term00)(\partial_{a b c}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b c}):
@substitute!(term00)(\partial_{a b}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b}):
@substitute!(term00)(\partial_{a}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a}):
@substitute!(term00)(\Gamma^{p}_{m n} -> Q^{p}_{m n}):

@substitute!(term01)(\partial_{a b c}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b c}):
@substitute!(term01)(\partial_{a b}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b}):
@substitute!(term01)(\partial_{a}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a}):
@substitute!(term01)(\Gamma^{p}_{m n} -> Q^{p}_{m n}):

# --- rebuild deriv05

deriv05:=@(term00) + @(term01):

# --- impose RNC conditions

@substitute!(deriv05)(Q^{p}_{m n} -> 0):
@substitute!(deriv05)(B^a B^b B^c Q^{p}_{a b c} -> 0):
@substitute!(deriv05)(B^a B^b B^c B^d Q^{p}_{a b c d} -> 0):
@substitute!(deriv05)(B^a B^b B^c B^d B^e Q^{p}_{a b c d e} -> 0);

# --- deriv06 to leading order in s equals  $-\Gamma^a_{bcdefg}$  which vanishes only in RNC

deriv06:=0;

```

$$deriv02 := -A^d Q^a{}_{bcd} B^b B^c - \frac{1}{2} A^e A^d Q^a{}_{bcd} B^b B^c - \frac{1}{6} A^f A^e A^d Q^a{}_{bcfed} B^b B^c - \frac{1}{24} A^g A^f A^e A^d Q^a{}_{bcg fed} B^b B^c$$

$$\begin{aligned}
deriv03 := & -A^f B^b B^c B^d Q^a_{cdfb} - \frac{1}{2} A^g A^f B^b B^c B^d Q^a_{cdgfb} + A^g A^f B^b B^c B^d Q^a_{bef} Q^e_{cdg} + A^g A^f B^b B^c B^d Q^a_{beg} Q^e_{cdf} \\
& + \frac{1}{3} A^h A^g A^f B^b B^c B^d Q^a_{begf} Q^e_{cdh} + \frac{1}{3} A^h A^g A^f B^b B^c B^d Q^a_{behf} Q^e_{cdg} + \frac{1}{3} A^h A^g A^f B^b B^c B^d Q^a_{bef} Q^e_{cdhg} \\
& + \frac{1}{3} A^h A^g A^f B^b B^c B^d Q^a_{behg} Q^e_{cdf} + \frac{1}{3} A^h A^g A^f B^b B^c B^d Q^a_{beg} Q^e_{cdhf} + \frac{1}{3} A^h A^g A^f B^b B^c B^d Q^a_{beh} Q^e_{cdgfh}
\end{aligned}$$

$$\begin{aligned}
deriv04 := & A^h B^b B^c B^d B^e Q^f_{bch} Q^a_{def} + 4 A^h B^b B^c B^d B^e Q^f_{bch} Q^a_{efd} + \frac{1}{2} A^i A^h B^b B^c B^d B^e Q^f_{bcih} Q^a_{def} + \frac{1}{2} A^i A^h B^b B^c B^d B^e Q^f_{bch} Q^a_{deif} \\
& + \frac{1}{2} A^i A^h B^b B^c B^d B^e Q^f_{bci} Q^a_{dehfh} + 2 A^i A^h B^b B^c B^d B^e Q^f_{bcih} Q^a_{efd} + 2 A^i A^h B^b B^c B^d B^e Q^f_{bch} Q^a_{efid} \\
& + 2 A^i A^h B^b B^c B^d B^e Q^f_{bci} Q^a_{efhd} + A^i A^h B^b B^c B^d B^e Q^a_{bfh} Q^f_{deic} + A^i A^h B^b B^c B^d B^e Q^a_{bfi} Q^f_{dehc}
\end{aligned}$$

$$deriv05 := A^j B^b B^c B^d B^e B^f Q^a_{bcg} Q^g_{efjd} + 3 A^j B^b B^c B^d B^e B^f Q^g_{bcj} Q^a_{efdg} + 6 A^j B^b B^c B^d B^e B^f Q^a_{cgb} Q^g_{efjd} + 6 A^j B^b B^c B^d B^e B^f Q^g_{bcj} Q^a_{fgde}$$

$$deriv06 := 0$$



# --- print the results -----

```
@print["x^a="~@(deriv00)];
@print["\frac{dx^a}{ds}="~@(deriv01)];
@print["\frac{d^2x^a}{ds^2}="~@(deriv02)~"+\BigO{\eps^6}"];
@print["\frac{d^3x^a}{ds^3}="~@(deriv03)~"+\BigO{\eps^6}"];
@print["\frac{d^4x^a}{ds^4}="~@(deriv04)~"+\BigO{\eps^6}"];
@print["\frac{d^5x^a}{ds^5}="~@(deriv05)~"+\BigO{\eps^6}"];
@print["\frac{d^6x^a}{ds^6}="~@(deriv06)~"+\BigO{\eps^6}"];
```

$$x^a = A^a$$

$$\frac{dx^a}{ds} = B^a$$

$$\frac{d^2x^a}{ds^2} = \left( -A^d Q^a{}_{bcd} B^b B^c - \frac{1}{2} A^e A^d Q^a{}_{bcd} B^b B^c - \frac{1}{6} A^f A^e A^d Q^a{}_{bcfed} B^b B^c - \frac{1}{24} A^g A^f A^e A^d Q^a{}_{bcgfed} B^b B^c \right) + \mathcal{O}(\epsilon^6)$$

$$\begin{aligned} \frac{d^3x^a}{ds^3} = & \left( -A^f B^b B^c B^d Q^a{}_{cdfb} - \frac{1}{2} A^g A^f B^b B^c B^d Q^a{}_{cdgfb} + A^g A^f B^b B^c B^d Q^a{}_{bef} Q^e{}_{cdg} + A^g A^f B^b B^c B^d Q^a{}_{beg} Q^e{}_{cdf} + \frac{1}{3} A^h A^g A^f B^b B^c B^d Q^a{}_{begf} Q^e{}_{cdh} \right. \\ & + \frac{1}{3} A^h A^g A^f B^b B^c B^d Q^a{}_{behf} Q^e{}_{cdg} + \frac{1}{3} A^h A^g A^f B^b B^c B^d Q^a{}_{bef} Q^e{}_{cdhg} + \frac{1}{3} A^h A^g A^f B^b B^c B^d Q^a{}_{behg} Q^e{}_{cdf} + \frac{1}{3} A^h A^g A^f B^b B^c B^d Q^a{}_{beg} Q^e{}_{cdhf} \\ & \left. + \frac{1}{3} A^h A^g A^f B^b B^c B^d Q^a{}_{beh} Q^e{}_{cdgf} \right) + \mathcal{O}(\epsilon^6) \end{aligned}$$

$$\begin{aligned} \frac{d^4x^a}{ds^4} = & \left( A^h B^b B^c B^d B^e Q^f{}_{bch} Q^a{}_{def} + 4 A^h B^b B^c B^d B^e Q^f{}_{bch} Q^a{}_{efd} + \frac{1}{2} A^i A^h B^b B^c B^d B^e Q^f{}_{bcih} Q^a{}_{def} + \frac{1}{2} A^i A^h B^b B^c B^d B^e Q^f{}_{bch} Q^a{}_{deif} \right. \\ & + \frac{1}{2} A^i A^h B^b B^c B^d B^e Q^f{}_{bci} Q^a{}_{dehf} + 2 A^i A^h B^b B^c B^d B^e Q^f{}_{bcih} Q^a{}_{efd} + 2 A^i A^h B^b B^c B^d B^e Q^f{}_{bch} Q^a{}_{efid} + 2 A^i A^h B^b B^c B^d B^e Q^f{}_{bci} Q^a{}_{efhd} \\ & \left. + A^i A^h B^b B^c B^d B^e Q^a{}_{bfh} Q^f{}_{deic} + A^i A^h B^b B^c B^d B^e Q^a{}_{bfi} Q^f{}_{dehc} \right) + \mathcal{O}(\epsilon^6) \end{aligned}$$

$$\frac{d^5x^a}{ds^5} = (A^j B^b B^c B^d B^e B^f Q^a{}_{bcg} Q^g{}_{efjd} + 3 A^j B^b B^c B^d B^e B^f Q^g{}_{bcj} Q^a{}_{efdg} + 6 A^j B^b B^c B^d B^e B^f Q^a{}_{cgb} Q^g{}_{efjd} + 6 A^j B^b B^c B^d B^e B^f Q^g{}_{bcj} Q^a{}_{fgde}) + \mathcal{O}(\epsilon^6)$$

$$\frac{d^6x^a}{ds^6} = 0 + \mathcal{O}(\epsilon^6)$$

```

# --- imported from connection.lib -----
# gamma(x) = Gamma00 + x^d Gamma01 + x^d x^e Gamma02 + x^d x^e x^f Gamma03 + ...
# Gamma01 = 1! gamma_{,d}      at x=0, symmetrised over {d}
# Gamma02 = 2! gamma_{,de}     at x=0, symmetrised over {d,e} = ( de + ed )/2!
# Gamma03 = 3! gamma_{,def}    at x=0, symmetrised over {d,e,f} = ( def + edf + dfe + ...)/3!

Gamma01:="import connection.lib Gamma01":
@run(Gamma01){"/Users/leo/local/sh/cdbfile"}:

Gamma02:="import connection.lib Gamma02":
@run(Gamma02){"/Users/leo/local/sh/cdbfile"}:

Gamma03:="import connection.lib Gamma03":
@run(Gamma03){"/Users/leo/local/sh/cdbfile"}:

Gamma04:="import connection.lib Gamma04":
@run(Gamma04){"/Users/leo/local/sh/cdbfile"}:

# --- Gamma0* as imported from connection.lib was *not* symmetrised, do so now -----
@sym!(Gamma02){ _{d}, _{e} }: @canonicalise!(%):
@sym!(Gamma03){ _{d}, _{e}, _{f} }: @canonicalise!(%):
@sym!(Gamma04){ _{d}, _{e}, _{f}, _{g} }: @canonicalise!(%):

# --- cancel the 1/n! that comes from @sym(...) -----
tmp:=@(Gamma02): Gamma02:= 2 @(tmp):
tmp:=@(Gamma03): Gamma03:= 6 @(tmp):
tmp:=@(Gamma04): Gamma04:= 24 @(tmp):

# --- substitute into the deriv0* -----
@substitute!(deriv02)(Q^{a}_{ }_{b c d} -> @(Gamma01)):
@substitute!(deriv02)(Q^{a}_{ }_{b c d e} -> @(Gamma02)):
@substitute!(deriv02)(Q^{a}_{ }_{b c d e f} -> @(Gamma03)):
@substitute!(deriv02)(Q^{a}_{ }_{b c d e f g} -> @(Gamma04)):
@distribute!(%): @prodsort!(%): @rename_dummies!(%): @canonicalise!(%):

@substitute!(deriv03)(Q^{a}_{ }_{b c d} -> @(Gamma01)):
@substitute!(deriv03)(Q^{a}_{ }_{b c d e} -> @(Gamma02)):

```

```

@substitute!(deriv03)(Q^{a}_{b c d e f} -> @(Gamma03)):
@distribute!(%): @prodsort!(%): @rename_dummies!(%): @canonicalise!(%):

@substitute!(deriv04)(Q^{a}_{b c d} -> @(Gamma01)):
@substitute!(deriv04)(Q^{a}_{b c d e} -> @(Gamma02)):
@substitute!(deriv04)(Q^{a}_{b c d e f} -> @(Gamma03)):
@distribute!(%): @prodsort!(%): @rename_dummies!(%): @canonicalise!(%):

@substitute!(deriv05)(Q^{a}_{b c d} -> @(Gamma01)):
@substitute!(deriv05)(Q^{a}_{b c d e} -> @(Gamma02)):
@substitute!(deriv05)(Q^{a}_{b c d e f} -> @(Gamma03)):
@distribute!(%): @prodsort!(%): @rename_dummies!(%): @canonicalise!(%):

@substitute!(deriv06)(Q^{a}_{b c d} -> @(Gamma01)):
@substitute!(deriv06)(Q^{a}_{b c d e} -> @(Gamma02)):
@substitute!(deriv06)(Q^{a}_{b c d e f} -> @(Gamma03)):
@distribute!(%): @prodsort!(%): @rename_dummies!(%): @canonicalise!(%):

# =====
#   force all indices on R to be downstairs
# =====

@substitute!(deriv02)(R^{a}_{b c d} -> R_{a b c d},
    R_{a}^{b}_{c d} -> R_{a b c d},
    R^{a}_{b c}^{d} -> R_{a b c d},
    R_{a b}^{c}_{d} -> R_{a b c d},
    R_{a b c}^{d} -> R_{a b c d},
    R^{a}_{b}^{c}_{d} -> R_{a b c d},
    R^{a}_{b}^{c}_{d} -> R_{a b c d},
    \nabla^{a}_{R_{b c d e}} -> \nabla_{a}{R_{b c d e}},
    \nabla^{a}_{f}{R_{b c d e}} -> \nabla_{a f}{R_{b c d e}},
    \nabla^{a}_{f g}{R_{b c d e}} -> \nabla_{a f g}{R_{b c d e}}):

@substitute!(deriv03)(R^{a}_{b c d} -> R_{a b c d},
    R_{a}^{b}_{c d} -> R_{a b c d},
    R^{a}_{b c}^{d} -> R_{a b c d},
    R_{a b}^{c}_{d} -> R_{a b c d},
    R_{a b c}^{d} -> R_{a b c d},
    R^{a}_{b}^{c}_{d} -> R_{a b c d},

```

```

R^{a}_{b}^{c}_{d} -> R_{a b c d},
\nabla^{a}_{R_{b c d e}} -> \nabla_{a}_{R_{b c d e}},
\nabla^{a}_{f}_{R_{b c d e}} -> \nabla_{a f}_{R_{b c d e}},
\nabla^{a}_{f g}_{R_{b c d e}} -> \nabla_{a f g}_{R_{b c d e}}):

@substitute!(deriv04)(R^{a}_{b c d} -> R_{a b c d},
R_{a}^{b}_{c d} -> R_{a b c d},
R^{a}_{b c}^{d} -> R_{a b c d},
R_{a b}^{c}_{d} -> R_{a b c d},
R_{a b c}^{d} -> R_{a b c d},
R^{a}_{b}^{c}_{d} -> R_{a b c d},
R^{a}_{b}^{c}_{d} -> R_{a b c d},
\nabla^{a}_{R_{b c d e}} -> \nabla_{a}_{R_{b c d e}},
\nabla^{a}_{f}_{R_{b c d e}} -> \nabla_{a f}_{R_{b c d e}},
\nabla^{a}_{f g}_{R_{b c d e}} -> \nabla_{a f g}_{R_{b c d e}}):

@substitute!(deriv05)(R^{a}_{b c d} -> R_{a b c d},
R_{a}^{b}_{c d} -> R_{a b c d},
R^{a}_{b c}^{d} -> R_{a b c d},
R_{a b}^{c}_{d} -> R_{a b c d},
R_{a b c}^{d} -> R_{a b c d},
R^{a}_{b}^{c}_{d} -> R_{a b c d},
R^{a}_{b}^{c}_{d} -> R_{a b c d},
\nabla^{a}_{R_{b c d e}} -> \nabla_{a}_{R_{b c d e}},
\nabla^{a}_{f}_{R_{b c d e}} -> \nabla_{a f}_{R_{b c d e}},
\nabla^{a}_{f g}_{R_{b c d e}} -> \nabla_{a f g}_{R_{b c d e}}):

@substitute!(deriv06)(R^{a}_{b c d} -> R_{a b c d},
R_{a}^{b}_{c d} -> R_{a b c d},
R^{a}_{b c}^{d} -> R_{a b c d},
R_{a b}^{c}_{d} -> R_{a b c d},
R_{a b c}^{d} -> R_{a b c d},
R^{a}_{b}^{c}_{d} -> R_{a b c d},
R^{a}_{b}^{c}_{d} -> R_{a b c d},
\nabla^{a}_{R_{b c d e}} -> \nabla_{a}_{R_{b c d e}},
\nabla^{a}_{f}_{R_{b c d e}} -> \nabla_{a f}_{R_{b c d e}},
\nabla^{a}_{f g}_{R_{b c d e}} -> \nabla_{a f g}_{R_{b c d e}}):

```

```

# --- compute the generalised connections -----
{B^{a},x^{a},R_{a b c d},R^{a}_{b c d},\nabla_{a}\{R_{b c d e}\}}::SortOrder.

@prodsort!(deriv02):
@prodsort!(deriv03):
@prodsort!(deriv04):
@prodsort!(deriv05):
@prodsort!(deriv06):

@rename_dummies!(deriv02):
@rename_dummies!(deriv03):
@rename_dummies!(deriv04):
@rename_dummies!(deriv05):
@rename_dummies!(deriv06):

@canonicalise!(deriv02):
@canonicalise!(deriv03):
@canonicalise!(deriv04):
@canonicalise!(deriv05):
@canonicalise!(deriv06):

# --- raise index {a} and set A^a -> x^a

genGamma02:= - @(deriv02)) g^{a u1}:
@distribute!(%): @eliminate_metric!(%):
tmp:=@(genGamma02) \delta^{a}_{u1}: @distribute!(%): @eliminate_kr!(%):
@substitute!%(A^{a}->x^{a}): @prodsort!(%): @rename_dummies!(%): @substitute!%(B^a -> 1):
genGamma02:= @(tmp); "genGamma02.trn"

genGamma03:= - @(deriv03)) g^{a u1}:
@distribute!(%): @eliminate_metric!(%):
tmp:=@(genGamma03) \delta^{a}_{u1}: @distribute!(%): @eliminate_kr!(%):
@substitute!%(A^{a}->x^{a}): @prodsort!(%): @rename_dummies!(%): @substitute!%(B^a -> 1):
genGamma03:= @(tmp); "genGamma03.trn"

genGamma04:= - @(deriv04) g^{a u1}:
@distribute!(%): @eliminate_metric!(%):
tmp:=@(genGamma04) \delta^{a}_{u1}: @distribute!(%): @eliminate_kr!(%):

```

```
@substitute! (%) (A^{a} -> x^{a}): @prodsort! (%): @rename_dummies! (%): @substitute! (%) (B^a -> 1):
genGamma04:= @(tmp); "genGamma04.trn"
```

```
genGamma05:= - @(deriv05)) g^{a u1}:
@distribute! (%): @eliminate_metric! (%):
tmp:=@(genGamma05) \delta^{a}_{u1}: @distribute! (%): @eliminate_kr! (%):
@substitute! (%) (A^{a} -> x^{a}): @prodsort! (%): @rename_dummies! (%): @substitute! (%) (B^a -> 1):
genGamma05:= @(tmp); "genGamma05.trn"
```

```
genGamma06:= - @(deriv06)) g^{a u1}:
@distribute! (%): @eliminate_metric! (%):
tmp:=@(genGamma06) \delta^{a}_{u1}: @distribute! (%): @eliminate_kr! (%):
@substitute! (%) (A^{a} -> x^{a}): @prodsort! (%): @rename_dummies! (%): @substitute! (%) (B^a -> 1):
genGamma06:= @(tmp); "genGamma06.trn"
```

```
@print["\Btag{11}\genGammaA=\killL"~@(genGamma02)~"\killR+\BigO{\eps^6}\Etag{11}"];
@print["\Btag{12}\genGammaB=\killL"~@(genGamma03)~"\killR+\BigO{\eps^6}\Etag{12}"];
@print["\Btag{13}\genGammaC=\killL"~@(genGamma04)~"\killR+\BigO{\eps^6}\Etag{13}"];
@print["\Btag{14}\genGammaD=\killL"~@(genGamma05)~"\killR+\BigO{\eps^6}\Etag{14}"];
@print["\Btag{15}\genGammaE=\killL"~@(genGamma06)~"\killR\Etag{15}"];
```

# --- the generalised connections in RNC coordinates -----

$$\begin{aligned}
\Gamma_{(bc)}^a(x) = & \frac{2}{3} x^d R^a{}_{bdc} + \frac{1}{6} x^d x^e \nabla_b R^a{}_{dec} + \frac{1}{3} x^d x^e \nabla_d R^a{}_{bec} + \frac{1}{12} x^d x^e \nabla^a R_{dbec} + \frac{8}{45} x^d x^e x^f R^a{}_{deg} R_{fbcg} - \frac{4}{45} x^d x^e x^f R^a{}_{bdg} R_{ecfg} - \frac{2}{45} x^d x^e x^f R^a{}_{dbg} R_{ecfg} \\
& + \frac{1}{10} x^d x^e x^f \nabla_{db} R^a{}_{efc} + \frac{1}{10} x^d x^e x^f \nabla_{de} R^a{}_{bfc} - \frac{2}{45} x^d x^e x^f R^a{}_{gdb} R_{ecfg} + \frac{1}{20} x^d x^e x^f \nabla_d R_{ebfc} + \frac{4}{45} x^d x^e x^f x^g R_{dbch} \nabla_e R^a{}_{fgh} \\
& + \frac{1}{30} x^d x^e x^f x^g R^a{}_{deh} \nabla_b R_{fcgh} + \frac{4}{45} x^d x^e x^f x^g R^a{}_{deh} \nabla_f R_{gbch} - \frac{2}{45} x^d x^e x^f x^g R^a{}_{bdh} \nabla_e R_{fcgh} - \frac{1}{45} x^d x^e x^f x^g R^a{}_{dbh} \nabla_e R_{fcgh} - \frac{1}{45} x^d x^e x^f x^g R_{dbeh} \nabla_c R^a{}_{fgh} \\
& - \frac{2}{45} x^d x^e x^f x^g R_{dbeh} \nabla_f R^a{}_{cgh} - \frac{1}{45} x^d x^e x^f x^g R_{dbeh} \nabla_f R^a{}_{gch} + \frac{1}{30} x^d x^e x^f x^g \nabla_{deb} R^a{}_{fgc} + \frac{1}{45} x^d x^e x^f x^g \nabla_{def} R^a{}_{bgc} - \frac{1}{36} x^d x^e x^f x^g R^a{}_{deh} \nabla_h R_{fbgc} \\
& - \frac{1}{45} x^d x^e x^f x^g R^a{}_{hdb} \nabla_e R_{fcgh} - \frac{1}{45} x^d x^e x^f x^g R_{dbeh} \nabla^a R_{fcgh} - \frac{1}{45} x^d x^e x^f x^g R_{dbeh} \nabla_f R^a{}_{hgc} + \frac{1}{60} x^d x^e x^f x^g \nabla_{de} R_{fbgc} + \mathcal{O}(\epsilon^6)
\end{aligned}$$

$$\begin{aligned}
\Gamma_{(bcd)}^a(x) = & \frac{1}{2} x^e \nabla_b R^a{}_{ced} + \frac{2}{15} x^e x^f R^a{}_{ebg} R_{fcdg} + \frac{8}{15} x^e x^f R^a{}_{beg} R_{fcdg} - \frac{2}{15} x^e x^f R^a{}_{bcg} R_{edfg} + \frac{3}{10} x^e x^f \nabla_{eb} R^a{}_{cfd} + \frac{1}{10} x^e x^f \nabla_{bc} R^a{}_{efd} \\
& + \frac{2}{5} x^e x^f R^a{}_{geb} R_{fcdg} + \frac{1}{20} x^e x^f \nabla_b R_{ecfd} + \frac{1}{9} x^e x^f x^g R_{ebch} \nabla_d R^a{}_{fgh} + \frac{1}{9} x^e x^f x^g R_{ebch} \nabla_h R^a{}_{fgd} + \frac{2}{9} x^e x^f x^g R_{ebch} \nabla_f R^a{}_{dgh} \\
& + \frac{2}{9} x^e x^f x^g R_{ebch} \nabla_f R^a{}_{hgd} + \frac{1}{9} x^e x^f x^g R_{ebch} \nabla^a R_{fdgh} + \frac{1}{9} x^e x^f x^g R^a{}_{beh} \nabla_c R_{fdgh} + \frac{2}{9} x^e x^f x^g R^a{}_{beh} \nabla_f R_{gcdh} \\
& - \frac{1}{18} x^e x^f x^g R^a{}_{beh} \nabla_h R_{fcgd} + \frac{1}{9} x^e x^f x^g R^a{}_{heb} \nabla_c R_{fdgh} + \frac{2}{9} x^e x^f x^g R^a{}_{heb} \nabla_f R_{gcdh} - \frac{1}{18} x^e x^f x^g R^a{}_{heb} \nabla_h R_{fcgd} + \mathcal{O}(\epsilon^6)
\end{aligned}$$

$$\begin{aligned}
\Gamma_{(bcde)}^a(x) = & \frac{4}{9} x^f R^a{}_{bcg} R_{fdeg} + \frac{1}{9} x^f x^g R^a{}_{bch} \nabla_d R_{fegh} + \frac{2}{9} x^f x^g R^a{}_{bch} \nabla_f R_{gdeh} - \frac{1}{18} x^f x^g R^a{}_{bch} \nabla_h R_{fdge} + \frac{1}{9} x^f x^g R_{fbch} \nabla_d R^a{}_{geh} + \frac{5}{9} x^f x^g R_{fbch} \nabla_d R^a{}_{hge} \\
& + \frac{2}{9} x^f x^g R_{fbch} \nabla_g R^a{}_{deh} + \frac{4}{9} x^f x^g R_{fbch} \nabla_h R^a{}_{dge} + \frac{1}{9} x^f x^g R_{fbch} \nabla^a R_{gdeh} + \frac{2}{3} x^f x^g R_{fbch} \nabla_d R^a{}_{egh} + \frac{1}{3} x^f x^g R^a{}_{bfh} \nabla_c R_{gdeh} + \frac{1}{3} x^f x^g R^a{}_{hfb} \nabla_c R_{gdeh} + \mathcal{O}(\epsilon^6)
\end{aligned}$$

$$\Gamma_{(bcdef)}^a(x) = \frac{2}{3} x^g R^a{}_{bch} \nabla_d R_{gef h} + x^g R_{gbch} \nabla_d R^a{}_{ef h} + \mathcal{O}(\epsilon^6)$$

$$\Gamma_{(bcdefg)}^a(x) = 0$$

```
# =====
#   export genGamma for use by truncate.cdbp
# =====

com:="open gen-gamma.lib":
@run(com){"/Users/leo/local/sh/cdbfile"}:

com:="export gen-gamma.lib genx2rnc.trn":
@run(com){"/Users/leo/local/sh/cdbfile"}:

com:="export gen-gamma.lib genGamma02.trn":
@run(com){"/Users/leo/local/sh/cdbfile"}:

com:="export gen-gamma.lib genGamma03.trn":
@run(com){"/Users/leo/local/sh/cdbfile"}:

com:="export gen-gamma.lib genGamma04.trn":
@run(com){"/Users/leo/local/sh/cdbfile"}:

com:="export gen-gamma.lib genGamma05.trn":
@run(com){"/Users/leo/local/sh/cdbfile"}:

com:="export gen-gamma.lib genGamma06.trn":
@run(com){"/Users/leo/local/sh/cdbfile"}:
```