
First we compute the covariant derivatives of Ra
bcd in terms of its partial derivatives.

In the second part of this notebook we will compute the partial derivatives in terms of the covariant derivatives.

We find a curious result. For the first four derivatives the results are identical under the following changes : ∂ → ∇ (on the R’s only) and Γ → −Γ. For the fifth
derivative the terms of the form R.dGamma.dˆ2Gamma are not equivalent under this change. You can spot these terms, there are just two of them and the have the
numerical coefficient 6.

The computations for the first four derivatives are quick – about 15 sec. But adding in just one more derivative, the fifth derivative, blows this out to about 10 mins.

# =================================================================================================

::KeepHistory(false).
::PostDefaultRules( @@collect_terms!(%) ).

{a,b,c,d,e,f,g,h,i,j,k,l,m,n,o,p,q,r,s,t,u,v,w#}::Indices.

# we can get away with using PartialDerivative for \nabla because we are computing
# symmetrised derivatives, this makes Cadabra’s job much easier

\nabla{#}::PartialDerivative.
\partial{#}::PartialDerivative.

R^{a}_{b c d}::RiemannTensor.

\delta^{a}_{b}::KroneckerDelta.

A^{a}::Depends(\partial).
B^{a u}_{b v}::Depends(\partial).
R^{a}_{b c d}::Depends(\partial).



\Gamma^{a}_{b c}::Depends(\partial).
\Gamma^{a}_{b c}::TableauSymmetry(shape={2}, indices={1,2}).

# --- construct the first four symmetrised partial derivatives of the Riemann tensor --------------

cderiv00:=R^{a}_{b c d} A^{b} A^{c} B^{d u}_{a v}:

cderiv01:=A^{c}\partial_{c}{@(cderiv00)}:
@distribute!(%):
@prodrule!(%):
@distribute!(%):
@substitute!(%)(A^{a}\partial_{a}{A^{b}} -> 0):
@substitute!(%)(A^{a}\partial_{a}{B^{d u}_{c v}} -> \Gamma^{e}_{c a}B^{d u}_{e v}A^{a}

-\Gamma^{d}_{e a}B^{e u}_{c v}A^{a}):
@distribute!(%):
@prodsort!(%):
@rename_dummies!(%):

cderiv02:=A^{c}\partial_{c}{@(cderiv01)}:
@distribute!(%):
@prodrule!(%):
@distribute!(%):
@substitute!(%)(A^{a}\partial_{a}{A^{b}} -> 0):
@substitute!(%)(A^{a}\partial_{a}{B^{d u}_{c v}} -> \Gamma^{e}_{c a}B^{d u}_{e v}A^{a}

-\Gamma^{d}_{e a}B^{e u}_{c v}A^{a}):
@distribute!(%):
@prodsort!(%):
@rename_dummies!(%):

cderiv03:=A^{c}\partial_{c}{@(cderiv02)}:
@distribute!(%):
@prodrule!(%):
@distribute!(%):
@substitute!(%)(A^{a}\partial_{a}{A^{b}} -> 0):
@substitute!(%)(A^{a}\partial_{a}{B^{d u}_{c v}} -> \Gamma^{e}_{c a}B^{d u}_{e v}A^{a}

-\Gamma^{d}_{e a}B^{e u}_{c v}A^{a}):
@distribute!(%):
@prodsort!(%):
@rename_dummies!(%):

cderiv04:=A^{c}\partial_{c}{@(cderiv03)}:
@distribute!(%):
@prodrule!(%):
@distribute!(%):
@substitute!(%)(A^{a}\partial_{a}{A^{b}} -> 0):
@substitute!(%)(A^{a}\partial_{a}{B^{d u}_{c v}} -> \Gamma^{e}_{c a}B^{d u}_{e v}A^{a}

-\Gamma^{d}_{e a}B^{e u}_{c v}A^{a}):
@distribute!(%):
@prodsort!(%):
@rename_dummies!(%):

cderiv05:=A^{c}\partial_{c}{@(cderiv04)}:
@distribute!(%):
@prodrule!(%):
@distribute!(%):
@substitute!(%)(A^{a}\partial_{a}{A^{b}} -> 0):
@substitute!(%)(A^{a}\partial_{a}{B^{d u}_{c v}} -> \Gamma^{e}_{c a}B^{d u}_{e v}A^{a}

-\Gamma^{d}_{e a}B^{e u}_{c v}A^{a}):
@distribute!(%):
@prodsort!(%):
@rename_dummies!(%):

# --- now turn to Riemann normal coordinates, defined by 0 = Γa
(bc,a) --------------

# --- temporarily replace all derivatives of \Gamma with derivatives of T -------------------------

@substitute!(cderiv02)(\partial_{d}{\Gamma^{a}_{b c}} -> T_{d}^{a}_{b c}):
@substitute!(cderiv03)(\partial_{d}{\Gamma^{a}_{b c}} -> T_{d}^{a}_{b c}):
@substitute!(cderiv04)(\partial_{d}{\Gamma^{a}_{b c}} -> T_{d}^{a}_{b c}):
@substitute!(cderiv05)(\partial_{d}{\Gamma^{a}_{b c}} -> T_{d}^{a}_{b c}):

@substitute!(cderiv03)(\partial_{d e}{\Gamma^{a}_{b c}} -> T_{d e}^{a}_{b c}):
@substitute!(cderiv04)(\partial_{d e}{\Gamma^{a}_{b c}} -> T_{d e}^{a}_{b c}):
@substitute!(cderiv05)(\partial_{d e}{\Gamma^{a}_{b c}} -> T_{d e}^{a}_{b c}):

@substitute!(cderiv04)(\partial_{d e f}{\Gamma^{a}_{b c}} -> T_{d e f}^{a}_{b c}):
@substitute!(cderiv05)(\partial_{d e f}{\Gamma^{a}_{b c}} -> T_{d e f}^{a}_{b c}):

@substitute!(cderiv05)(\partial_{d e f g}{\Gamma^{a}_{b c}} -> T_{d e f g}^{a}_{b c}):

# --- now set \Gamma to zero ----------------------------------------------------------------------

@substitute!!(cderiv01)(\Gamma^{a}_{b c} -> 0):
@substitute!!(cderiv02)(\Gamma^{a}_{b c} -> 0):
@substitute!!(cderiv03)(\Gamma^{a}_{b c} -> 0):
@substitute!!(cderiv04)(\Gamma^{a}_{b c} -> 0):
@substitute!!(cderiv05)(\Gamma^{a}_{b c} -> 0):

# --- now re-introduce the derivatives of \Gamma --------------------------------------------------

@substitute!(cderiv02)(T_{d}^{a}_{b c} -> \partial_{d}{\Gamma^{a}_{b c}}):
@substitute!(cderiv03)(T_{d}^{a}_{b c} -> \partial_{d}{\Gamma^{a}_{b c}}):
@substitute!(cderiv04)(T_{d}^{a}_{b c} -> \partial_{d}{\Gamma^{a}_{b c}}):
@substitute!(cderiv05)(T_{d}^{a}_{b c} -> \partial_{d}{\Gamma^{a}_{b c}}):

@substitute!(cderiv03)(T_{d e}^{a}_{b c} -> \partial_{d e}{\Gamma^{a}_{b c}}):
@substitute!(cderiv04)(T_{d e}^{a}_{b c} -> \partial_{d e}{\Gamma^{a}_{b c}}):
@substitute!(cderiv05)(T_{d e}^{a}_{b c} -> \partial_{d e}{\Gamma^{a}_{b c}}):

@substitute!(cderiv04)(T_{d e f}^{a}_{b c} -> \partial_{d e f}{\Gamma^{a}_{b c}}):
@substitute!(cderiv05)(T_{d e f}^{a}_{b c} -> \partial_{d e f}{\Gamma^{a}_{b c}}):

@substitute!(cderiv05)(T_{d e f g}^{a}_{b c} -> \partial_{d e f g}{\Gamma^{a}_{b c}}):

# --- done with B, so get rid of it ---------------------------------------------------------------

@substitute!(cderiv01)(B^{a u}_{b v} -> \delta^{a}_{v}\delta^{u}_{b}): @distribute!(%): @eliminate_kr!(%):
@substitute!(cderiv02)(B^{a u}_{b v} -> \delta^{a}_{v}\delta^{u}_{b}): @distribute!(%): @eliminate_kr!(%):
@substitute!(cderiv03)(B^{a u}_{b v} -> \delta^{a}_{v}\delta^{u}_{b}): @distribute!(%): @eliminate_kr!(%):
@substitute!(cderiv04)(B^{a u}_{b v} -> \delta^{a}_{v}\delta^{u}_{b}): @distribute!(%): @eliminate_kr!(%):
@substitute!(cderiv05)(B^{a u}_{b v} -> \delta^{a}_{v}\delta^{u}_{b}): @distribute!(%): @eliminate_kr!(%):

# --- tidy up and display the results ------------------------------------------------------------

tmp01:=@(cderiv01): @canonicalise!(%): @factor_out!!(%){A^{a}}:
tmp02:=@(cderiv02): @canonicalise!(%): @factor_out!!(%){A^{a}}:
tmp03:=@(cderiv03): @canonicalise!(%): @factor_out!!(%){A^{a}}:
tmp04:=@(cderiv04): @canonicalise!(%): @factor_out!!(%){A^{a}}:
tmp05:=@(cderiv05): @canonicalise!(%): @factor_out!!(%){A^{a}}:

@substitute!(tmp01)(\partial_{p}{R^{a}_{b d c}} -> - \partial_{p}{R^{a}_{b c d}}):
@substitute!(tmp02)(\partial_{p q}{R^{a}_{b d c}} -> - \partial_{p q}{R^{a}_{b c d}}):
@substitute!(tmp03)(\partial_{p q r}{R^{a}_{b d c}} -> - \partial_{p q r}{R^{a}_{b c d}}):
@substitute!(tmp04)(\partial_{p q r s}{R^{a}_{b d c}} -> - \partial_{p q r s}{R^{a}_{b c d}}):
@substitute!(tmp05)(\partial_{p q r s t}{R^{a}_{b d c}} -> - \partial_{p q r s t}{R^{a}_{b c d}}):

@print["\cderivA="~@(tmp01)~"?"];
@print["\cderivB="~@(tmp02)~"?"];
@print["\cderivC="~@(tmp03)~"?"];
@print["\cderivD="~@(tmp04)~"?"];
@print["\cderivE="~@(tmp05)~"?"];

The first five symmetrised covariant derivatives of Ra
bcd in a Riemann normal coordinate frame.

Now we compute the partial derivatives of Ra
bcd in terms of its covariant derivatives.

We do this by swapping term across the equal sign in the previous result followed by substitutions of lower oder partial derivatives from previous results. Huh?

A typical result of the previous calculations will be of the form
Ra

(bcḋ;pq)
= Ra

(bcḋ,pq)
+ Qa

bcdpq

where Q contains lower oder partial derivatives of R (and various derivatives of the connections). We rewrite this as

Ra
(bcḋ,pq)

= Ra
(bcḋ;pq)

−Qa
bcdpq

The lower order partial derivatives of R that appear on the right hand side can be eliminated by substituitions from previous results.

# --- now re-arrange equations to give partial derivs in terms of covariant derivs ----------------

tmp01:= - @(cderiv01):
tmp02:= - @(cderiv02):
tmp03:= - @(cderiv03):
tmp04:= - @(cderiv04):
tmp05:= - @(cderiv05):

# --- replace highest order \partial with \nabla --------------------------------------------------

@substitute!(tmp01)(\partial_{p}{R^{a}_{b c d}} -> - \nabla_{p}{R^{a}_{b c d}}):
@substitute!(tmp02)(\partial_{p q}{R^{a}_{b c d}} -> - \nabla_{p q}{R^{a}_{b c d}}):
@substitute!(tmp03)(\partial_{p q r}{R^{a}_{b c d}} -> - \nabla_{p q r}{R^{a}_{b c d}}):
@substitute!(tmp04)(\partial_{p q r s}{R^{a}_{b c d}} -> - \nabla_{p q r s}{R^{a}_{b c d}}):
@substitute!(tmp05)(\partial_{p q r s t}{R^{a}_{b c d}} -> - \nabla_{p q r s t}{R^{a}_{b c d}}):

# --- use previous results to eliminate lower oder \partial’s -------------------------------------

@substitute!(tmp02)(A^{a} A^{b} A^{c}\partial_{c}{R^{u}_{a b v}} -> @(tmp01)):
@substitute!(tmp03)(A^{a} A^{b} A^{c}\partial_{c}{R^{u}_{a b v}} -> @(tmp01)):
@substitute!(tmp04)(A^{a} A^{b} A^{c}\partial_{c}{R^{u}_{a b v}} -> @(tmp01)):
@substitute!(tmp05)(A^{a} A^{b} A^{c}\partial_{c}{R^{u}_{a b v}} -> @(tmp01)):

@substitute!(tmp03)(A^{a} A^{b} A^{c} A^{d}\partial_{c d}{R^{u}_{a b v}} -> @(tmp02)):
@substitute!(tmp04)(A^{a} A^{b} A^{c} A^{d}\partial_{c d}{R^{u}_{a b v}} -> @(tmp02)):
@substitute!(tmp05)(A^{a} A^{b} A^{c} A^{d}\partial_{c d}{R^{u}_{a b v}} -> @(tmp02)):

@substitute!(tmp04)(A^{a} A^{b} A^{c} A^{d} A^{e}\partial_{c d e}{R^{u}_{a b v}} -> @(tmp03)):
@substitute!(tmp05)(A^{a} A^{b} A^{c} A^{d} A^{e}\partial_{c d e}{R^{u}_{a b v}} -> @(tmp03)):

@substitute!(tmp05)(A^{a} A^{b} A^{c} A^{d} A^{e} A^{f}\partial_{c d e f}{R^{u}_{a b v}} -> @(tmp04)):

@distribute!(tmp01):
@distribute!(tmp02):
@distribute!(tmp03):
@distribute!(tmp04):
@distribute!(tmp05):

# --- tidy up and display the results ------------------------------------------------------------

pderiv01:=@(tmp01): @prodsort!(%): @rename_dummies!(%): @canonicalise!(%): @factor_out!!(%){A^{a}}:
pderiv02:=@(tmp02): @prodsort!(%): @rename_dummies!(%): @canonicalise!(%): @factor_out!!(%){A^{a}}:
pderiv03:=@(tmp03): @prodsort!(%): @rename_dummies!(%): @canonicalise!(%): @factor_out!!(%){A^{a}}:
pderiv04:=@(tmp04): @prodsort!(%): @rename_dummies!(%): @canonicalise!(%): @factor_out!!(%){A^{a}}:
pderiv05:=@(tmp05): @prodsort!(%): @rename_dummies!(%): @canonicalise!(%): @factor_out!!(%){A^{a}}:

@substitute!(pderiv01)(\nabla_{p}{R^{a}_{b d c}} -> - \nabla_{p}{R^{a}_{b c d}}):
@substitute!(pderiv02)(\nabla_{p q}{R^{a}_{b d c}} -> - \nabla_{p q}{R^{a}_{b c d}}):
@substitute!(pderiv03)(\nabla_{p q r}{R^{a}_{b d c}} -> - \nabla_{p q r}{R^{a}_{b c d}}):
@substitute!(pderiv04)(\nabla_{p q r s}{R^{a}_{b d c}} -> - \nabla_{p q r s}{R^{a}_{b c d}}):
@substitute!(pderiv05)(\nabla_{p q r s t}{R^{a}_{b d c}} -> - \nabla_{p q r s t}{R^{a}_{b c d}}):

@print["\pderivA="~@(pderiv01)~"?"];
@print["\pderivB="~@(pderiv02)~"?"];
@print["\pderivC="~@(pderiv03)~"?"];
@print["\pderivD="~@(pderiv04)~"?"];
@print["\pderivE="~@(pderiv05)~"?"];

The first five symmetrised partial derivatives of Ra
bcd in a Riemann normal coordinate frame.

