
Covariant differentiation

One way to compute the covariant derivative va;b is to expand the right hand side of

va;bA
aDb =

d (vaAa)
ds

while using

dva

ds
= va,bD

b

0 = ∇D Aa =
dAa

ds
+ Γa

bcA
bDc

0 = ∇D Da =
dDa

ds
+ Γa

bcD
bDc

and then to compare corresponding coefficients of the AaDb terms. This leads directly to the standard result

va;b = va,b − Γc
abvc

# --- some basic declarations -------------------------------------------------

::PostDefaultRules( @@collect_terms!(%) ).

{a,b,c,d,e,f,g,h,i,j,k,l,m,n,o,p,q,r,s,t,u#}::Indices.

\nabla{#}::PartialDerivative.
\partial{#}::PartialDerivative.

A^{a}::Depends(\partial).
D^{a}::Depends(\partial).
v_{a}::Depends(\partial).
\Gamma^{a}_{b c}::Depends(\partial).

# --- compute the covariant derivative ---------------------------------------

rhs00:=v_{a} A^{a}:

rhs01:=D^{c}\partial_{c}{@(rhs00)}:
@distribute!(%):
@prodrule!(%):
@distribute!(%):
@substitute!(%)(D^{a}\partial_{a}{A^{b}} -> -\Gamma^{b}_{a c}A^{a}D^{c}):
@substitute!(%)(D^{a}\partial_{a}{D^{b}} -> -\Gamma^{b}_{a c}D^{a}D^{c}):
@prodsort!(%):
@rename_dummies!(%):
@canonicalise!(%):

# --- tidy up and display the results ----------------------------------------

@factor_out!!(rhs00){A^{a}}:
@factor_out!!(rhs00){D^{a}};

@factor_out!!(rhs01){A^{a}}:
@factor_out!!(rhs01){D^{a}};

# --- Cadabra output is printed only when a %! command is given ---------------

rhs00 := vaAa

rhs01 := AaDb (∂bva − Γc
abvc)



Symmetrised covariant derivatives

Repeated application of d/ds will produce symmetrised covariant derivatives,

va;i1i2i3···inAaDi1Di2Di3 · · ·Din =
dn (vaAa)

dsn

The procedure is similar to the previous example. We expand the right hand side, one derivative at a time,
while using the above equations to eliminate the ordinary derivatives in Aa and Da.

Now we shall train Cadabra to compute va;(bcd)A
aDbDcDd.

# --- some basic declarations -------------------------------------------------

\partial{#}::PartialDerivative.

A^{a}::Depends(\partial).
D^{a}::Depends(\partial).
v_{a}::Depends(\partial).
\Gamma^{a}_{b c}::Depends(\partial).

# --- compute the first 3 symmetrised covariant derivatives ------------------

rhs00:=v_{a} A^{a}:

rhs01:=D^{c}\partial_{c}{@(rhs00)}:
@distribute!(%):
@prodrule!(%):
@distribute!(%):
@substitute!(%)(D^{a}\partial_{a}{A^{b}} -> -\Gamma^{b}_{a c}A^{a}D^{c}):
@substitute!(%)(D^{a}\partial_{a}{D^{b}} -> -\Gamma^{b}_{a c}D^{a}D^{c}):
@prodsort!(%):
@rename_dummies!(%):
@canonicalise!(%):

rhs02:=D^{c}\partial_{c}{@(rhs01)}:
@distribute!(%):
@prodrule!(%):
@distribute!(%):
@substitute!(%)(D^{a}\partial_{a}{A^{b}} -> -\Gamma^{b}_{a c}A^{a}D^{c}):
@substitute!(%)(D^{a}\partial_{a}{D^{b}} -> -\Gamma^{b}_{a c}D^{a}D^{c}):
@prodsort!(%):
@rename_dummies!(%):
@canonicalise!(%):

rhs03:=D^{c}\partial_{c}{@(rhs02)}:
@distribute!(%):
@prodrule!(%):
@distribute!(%):
@substitute!(%)(D^{a}\partial_{a}{A^{b}} -> -\Gamma^{b}_{a c}A^{a}D^{c}):
@substitute!(%)(D^{a}\partial_{a}{D^{b}} -> -\Gamma^{b}_{a c}D^{a}D^{c}):
@prodsort!(%):
@rename_dummies!(%):
@canonicalise!(%):

# --- tidy up and display the results ----------------------------------------

@factor_out!!(rhs00){A^{a}}:
@factor_out!!(rhs00){D^{a}};

@factor_out!!(rhs01){A^{a}}:
@factor_out!!(rhs01){D^{a}};

@factor_out!!(rhs02){A^{a}}:
@factor_out!!(rhs02){D^{a}};

@factor_out!!(rhs03){A^{a}}:
@factor_out!!(rhs03){D^{a}};

rhs00 := vaAa

rhs01 := AaDb (∂bva − Γc
abvc)

rhs02 := AaDbDc
(
−2 Γd

ab∂cvd − Γd
bc∂dva + ∂bcva + Γd

abΓe
dcve + Γd

aeΓe
bcvd − ∂bΓd

acvd

)
rhs03 := AaDbDcDd

(
3 Γe

abΓf
ec∂dvf + 3 Γe

af Γf
bc∂dve + 3 Γe

abΓf
cd∂fve − 3 ∂bΓe

ac∂dve − 3 Γe
ab∂cdve

+ Γe
bcΓf

ed∂fva + Γe
bcΓf

de∂fva − ∂bΓe
cd∂eva − 3 Γe

bc∂deva + ∂bcdva − Γe
abΓf

ecΓg
fdvg

− Γe
af Γf

bcΓg
edvg − 2 Γe

abΓf
cdΓg

efvg + Γe
fb∂cΓf

adve + 2 Γe
ab∂cΓf

edvf − Γe
af Γf

gbΓg
cdve

− Γe
af Γf

bgΓg
cdve + 2 Γe

bc∂dΓf
aevf + Γe

af∂bΓf
cdve + Γe

bc∂eΓf
advf − ∂bcΓe

adve

)

Covariant derivatives and the Riemann tensor

# --- some basic declarations -------------------------------------------------

\partial{#}::PartialDerivative.

A^{a}::Depends(\partial).
D^{a}::Depends(\partial).
E^{a}::Depends(\partial).
v_{a}::Depends(\partial).
\Gamma^{a}_{b c}::Depends(\partial).
\Gamma^{a}_{b c}::TableauSymmetry(shape={2}, indices={1,2}).

# --- construct the scalar v_{a} A^{a} ---------------------------------------

scalar:=v_{a} A^{a}:

# --- compute the covariant derivative in the direction of D^a ---------------

derivD:=D^{c}\partial_{c}{@(scalar)}:
@distribute!(%):
@prodrule!(%):
@distribute!(%):
@substitute!(%)(D^{a}\partial_{a}{A^{b}} -> -\Gamma^{b}_{a c}A^{a}D^{c}):
@substitute!(%)(D^{a}\partial_{a}{D^{b}} -> -\Gamma^{b}_{a c}D^{a}D^{c}):
@prodsort!(%):
@rename_dummies!(%):
@canonicalise!(%):

# --- compute the covariant derivative in the direction of E^a ---------------

derivDE:=E^{c}\partial_{c}{@(derivD)}:
@distribute!(%):
@prodrule!(%):
@distribute!(%):
@substitute!(%)(E^{a}\partial_{a}{A^{b}} -> -\Gamma^{b}_{a c}A^{a}E^{c}):
@substitute!(%)(E^{a}\partial_{a}{D^{b}} -> -\Gamma^{b}_{a c}D^{a}E^{c}):
@substitute!(%)(E^{a}\partial_{a}{E^{b}} -> -\Gamma^{b}_{a c}E^{a}E^{c}):
@prodsort!(%):
@rename_dummies!(%):
@canonicalise!(%);

# --- copy to derivED then swap the order of the derivatives ------------------

derivED:=@(derivDE):

@substitute!(%)(E^{a} -> F^{a}):
@substitute!(%)(D^{a} -> E^{a}):
@substitute!(%)(F^{a} -> D^{a});

# --- subtract and simplify ---------------------------------------------------

diff:=@(derivDE) - @(derivED):

{A^{a},D^{a},E^{a},v_{a},\Gamma^{a}_{b c}}::SortOrder.

@prodsort!(%):
@rename_dummies!(%):
@canonicalise!(%):
@collect_terms!(%);

# --- tidy up and display the results ----------------------------------------

@factor_out!(%){A^{a}}:
@factor_out!(%){D^{a}}:
@factor_out!(%){E^{a}}:
@factor_out!(%){v_{a}};

# --- Cadabra output is printed only when a %! command is given ---------------

@print["A^a D^b E^c (v_{a;b;c}-v_{a;c;b})="~@(diff)];

derivDE := −AaDbEcΓd
ac∂bvd −AaDbEcΓd

bc∂dva + AaDbEc∂bcva + AaDbEcΓd
acΓe

bdve

+ AaDbEcΓd
aeΓe

bcvd −AaDbEc∂cΓd
abvd −AaDbEcΓd

ab∂cvd

derivED := −AaEbDcΓd
ac∂bvd −AaEbDcΓd

bc∂dva + AaEbDc∂bcva + AaEbDcΓd
acΓe

bdve

+ AaEbDcΓd
aeΓe

bcvd −AaEbDc∂cΓd
abvd −AaEbDcΓd

ab∂cvd

diff := AaDbEcvdΓd
beΓe

ac −AaDbEcvd∂cΓd
ab −AaDbEcvdΓd

ceΓe
ab + AaDbEcvd∂bΓd

ac

diff := AaDbEcvd

(
Γd

beΓe
ac − ∂cΓd

ab − Γd
ceΓe

ab + ∂bΓd
ac

)
AaDbEc (va;b;c − va;c;b) = AaDbEcvd

(
Γd

beΓe
ac − ∂cΓd

ab − Γd
ceΓe

ab + ∂bΓd
ac

)

Printing real equations

That’s all very nice, but how do I get real equations printed? (i.e. with both the left and right hand sides
looking like normal mathematics). Simply use Cadabra’s @print algorithm. Here is how we could could display
some of the results from the previous computation.

@print["v_{a;b}A^aD^b="~@(rhs01)];
@print["v_{a;bc}A^aD^bD^c="~@(rhs02)];

va;bA
aDb = AaDb (∂bva − Γc

abvc)

va;bcA
aDbDc = AaDbDc

(
−2 Γd

ab∂cvd − Γd
bc∂dva + ∂bcva + Γd

abΓe
dcve + Γd

aeΓe
bcvd − ∂bΓd

acvd

)

Libraries

Cadabra has a feature where the output of a single expression can be sent to a file. Here is an example.

rhs:=@(rhs02); "rhs.txt"

The above will save the output in the file rhs.txt. Can we save many expressions in one file? No, the above
syntax (as far as I can tell) limits you to one expression per file.

I’ve written a shell script and a support program that allow us to create libraries of useful cadabra expressions.
These libraries can be used by other cadabra notebooks. Here is an example where a new library is created and
two expressions are exported.

# --- create two new expressions and save their output

result01:=@(rhs01); "result01.txt"
result02:=@(rhs02); "result02.txt"

# --- create a library, this deletes any previous libray named sample.lib

com:="open sample.lib":
@run(com){"/Users/leo/local/sh/cdbfile"}:

# --- append result01.txt and result02.txt to the library and delete the .txt files

com:="export sample.lib result01.txt":
@run(com){"/Users/leo/local/sh/cdbfile"}:

com:="export sample.lib result02.txt":
@run(com){"/Users/leo/local/sh/cdbfile"}:

Note that when the export command finishes it will also delete the file it read from (for example result01.txt
and result02.txt). The symbol com must be created before the @run command as it forms the link between
Cadabra and the unix shell. It first contains the arguments to be passed to the shell script and later it receives
the shell script’s output.

You can also import expressions from other libraries

# --- import metric from metric.lib

metric:="import metric.lib metric":
@run(metric){"/Users/leo/local/sh/cdbfile"}:

# --- import the inverse metric from metric-inv.lib

inverse:="import metric-inv.lib invMetric":
@run(inverse){"/Users/leo/local/sh/cdbfile"};

Notice the change of name in the above example. The inverse metric was known as invMetric inside the library
metric-inv.lib but in this notebook we will use the name inverse. Just to prove that we have sucessfully
imported the expression, here it is

inverse := gab +
1
3

xcxdRa
c

b
d +

1
6

xcxdxe∇cR
a

d
b

e

Configuration

Each TeX file created by cdbmerge will begin with

\documentclass{cadabra}

You are welcome to cutsomise the cadabra.cls class to whatever suits your fancy. But such changes will effect
every TeX file (well just those that use the cdbmerge program). If you want to make changes that apply to

one file, say fred.tex, then you can create a file fred.cfg and place in it any set of LaTeX commands. The
cadabra class will instruct LaTeX to read this file (if it exists) as the last step in processing the \documentclass
command.

One useful macro that is defined in cadabra.cls is the \papersize macro. It takes six arguments being the
height, width, left right, top and bottom margin sizes. For exampe,

\papersize{210mm}{297mm}{2cm}{2cm}{2cm}{2cm}}

will produce an A4 paper in landsacpe form with 2cm margins along all four edges.

One limitation of the cdbmerge program is that it does not respect the cadabra attribute LaTeXForm. That
is, if you use

\pd::LaTeXForm{\partial}.

in your cadabra code you will find that LaTeX will complain about the undefined symbol \pd. This problem is
easily fixed by including

\def\pd{\partial}

in your configuration file (e.g. fred.cfg).

